

# Complément au manuel du Basic Alcyane

Le manuel du Basic Alcyane n'est pas totalement complet. Vous aurez certainement remarqué qu'il manque les pages 2.10 et 9.3. La page 2.10 correspond à l'instruction `KEEP` tandis que j'ignore absolument pourquoi la page 9.3 est absente.

Les pages qui suivent contiennent la description de l'instruction `KEEP` ainsi que des compléments sur les instructions du mode graphique.

Michel LUCAS – Octobre 2023

## KEEP

### FONCTION

Cette instruction permet de conserver une copie du basic présent en mémoire en spécifiant le programme à charger et exécuter lors du chargement de ce basic. Ceci évite le recours à une opération manuelle source d'erreurs potentielles.

### FORME GÉNÉRALE

```
KEEP (nd1) "BASIC" [/ (nd2) "PROG" [/RUN] ]
```

### RÈGLES

- nd1 est le numéro du disque sur lequel sera enregistré le basic présent en mémoire.
- "BASIC" est le nom sous lequel sera enregistré le basic présent en mémoire. Ce nom ne doit pas être déjà présent sur le disque nd1.
- Il est possible de spécifier optionnellement le nom du programme à charger lors du chargement de "BASIC". Il faut alors définir le disque sur lequel il se trouve (nd2) ainsi que son nom "PROG". Cette commande est équivalente à un LOAD#.
- Il est également possible de spécifier si ce programme doit être exécuté après son chargement en ajoutant /RUN à la fin de la ligne.

### EXEMPLES

Enregistrement du basic en mémoire sur l'unité 3 sous le nom « B3 ».

```
KEEP (3) "B3"
```

Enregistrement du basic en mémoire sur l'unité 9 sous le nom « AUTO » et lancement du programme « COMPTA » sur l'unité 10.

```
KEEP (9) "AUTO" / (10) "COMPTA" /RUN
```

## PLOT

### FONCTION

Cette instruction permet de tracer un vecteur, de poursuivre un tracé de vecteur, et de changer la couleur d'un vecteur.

### FORME GÉNÉRALE

```
PLOT Xd,Yd[,Xf,Yf[,C]]
```

### RÈGLES

- La forme générale permet de tracer un vecteur du point de coordonnées (Xd, Yd) au point de coordonnées (Xf, Yf), avec la couleur C.
- Les coordonnées doivent être comprises entre 0 et 255.
- La couleur peut varier de 0 à 15. Elle dépend du nombre de plans de mémoire intégrés dans la machine. Il faut 4 plans mémoire pour obtenir 16 couleurs.
- Le point de coordonnées (0, 0) est situé en bas à gauche de l'écran.
- Il est possible de tracer un vecteur partant de l'extrémité du dernier vecteur tracé en définissant seulement deux coordonnées et une couleur, par exemple `PLOT 100,200,1`.
- Si seule la couleur est définie, le dernier vecteur tracé est retracé avec la nouvelle couleur, par exemple `PLOT 0` pour effacer le dernier vecteur tracé.

### EXEMPLES

```
0010 REM TRACE DE RECTANGLE  
0020 PLOT 10,10,100,10,1  
0030 PLOT 100,200,1  
0040 PLOT 200,10,1  
0050 PLOT 10,10,1  
0060 END
```

## CIRCLE

### FONCTION

Cette fonction permet de tracer des cercles sur le plan graphique.

Remarque : le plan graphique de 255 x 255 pixels est affiché comme un rectangle sur l'écran d'Alcyane, ce qui déforme légèrement le cercle.

### FORME GÉNÉRALE

```
CIRCLE R, X, Y, C
```

### RÈGLES

- X et Y sont les coordonnées du centre du cercle. Leurs valeurs doivent être comprises entre 0 et 255.
- R est le rayon du cercle. Sa valeur doit être comprise entre 1 et 127.
- C est la couleur de tracé. Sa valeur doit être comprise entre 0 et 15.
- Une erreur 98 est signalée si l'un des paramètres a une valeur invalide.
- Aucun contrôle n'est effectué pour savoir si le tracé déborde de l'écran. Si cela se produit, la partie manquante est affichée dans la zone diamétralement opposée de l'écran.

### EXEMPLES

```
0010 TRACE D'UN DIAGRAMME PIE CHART
0020 RAYON=80:X=127:Y=127:C=0
0030 CIRCLE RAYON,X,Y,1
0040 FOR I=1 TO 16:REM NOMBRE DE PARTS
0050 READ B
0060 C=C+B: ANGLE=C*3.6:REM 360/100
0070 XE=Y+RAYON*COS ANGLE
0080 YE=Y+RAYON*SIN ANGLE
0090 PLOT X,Y,XE,YE,1
0100 NEXT I
0110 REM TABLEAU DES POURCENTAGES DE LA SURFACE
0120 DATA 10,20,35,15,16,4
0130 END
```

## FILL

### FONCTION

Cette fonction permet le remplissage de surfaces graphiques fermées.

### FORME GÉNÉRALE

FILL X, Y, C

### RÈGLES

- Les limites de la surface à remplir doivent être des courbes fermées.
- Il n'y a pas de limitation à la forme de la surface à remplir.
- X et Y sont les coordonnées du point de départ du remplissage. Ce point doit être à l'intérieur de la surface à remplir.
- Les coordonnées X et Y doivent être comprises entre 0 et 255.
- C est la couleur de remplissage. Elle doit être comprise entre 0 et 15.
- Une erreur dans l'un des paramètres déclenche une erreur 98.

### EXEMPLES

```
0010 TRACE D'UN LOSANGE PLEIN
0020 PLOT 20,10,10,25,1
0030 PLOT 20,40,1
0040 PLOT 30,25,1
0050 PLOT 20,10,1
0060 FILL 20,25,1
0070 END
```

## GPATTERN

### FONCTION

Cette commande permet de définir le dessin d'un vecteur sur 16 points. Elle permet, par exemple, de tracer des pointillés ou des traits d'axe au moyen de la commande PLOT.

### FORME GÉNÉRALE

GPATTERN Exp

### RÈGLES

- Exp est une expression dont le résultat doit être compris entre 0 et 65535.
- Le résultat de l'expression est une valeur décimale dont la représentation binaire sur 16 bits définit le dessin du vecteur. Par exemple, si le vecteur à tracer doit avoir un point sur deux, la valeur binaire de l'expression sera 1010101010101010 = 43690 en décimal.

### EXEMPLES

```
0010 REM CONVERSION BINAIRE -> DECIMAL
0020 DIM A$(16)
0030 INIT A$ TO "0"
0040 RES=0
0050 MODIF A$
0060 FOR I=1 TO 16
0070 RES=RES*2+(A$(I)#0)
0080 NEXT I
0090 DISP RES
0100 END

0010 REM DEFINITIONS DE TRAIT DIVERSES
0020 REM TRAIT D'AXE 1110010011100100
0030 GPATTERN 58596
0040 REM TRAIT INTERROMPU COURT 1100110011001100
0050 GPATTERN 52428
0060 REM TRAIT INTERROMPU LONG 1110111011101110
0070 GPATTERN 61166
0080 END
```

## GCOPY

### FONCTION

Cette commande permet de recopier l'écran graphique sur une imprimante.

### FORME GÉNÉRALE

GCOPY Per [, Yd, Yf, Ech]

### RÈGLES

- Per est le numéro de périphérique de sortie (4 ou 7 pour une imprimante).
- Yd est l'ordonnée de début de copie. Sa valeur doit être comprise entre 0 et 255.
- Yf est l'ordonnée de fin de copie. Sa valeur doit être comprise entre 0 et 255.
- L'ordonnée du haut de l'écran est 255. Yd doit être nécessairement supérieur à Yf.
- Ech est l'échelle du graphique imprimé. Elle peut prendre les valeurs 1 ou 2 uniquement.
- Un paramètre en défaut est signalé par une erreur 98.

### EXEMPLES

```
0010 REM RECOPIE DE TOUT L'ECRAN GRAPHIQUE SUR L'IMPRIMANTE
0020 GCOPY 4
0030 END
```

```
0010 REM RECOPIE DE LA MOITIE SUPERIEURE DU GRAPHIQUE
0020 REM SUR L'IMPRIMANTE A L'ECHELLE 2
0030 GCOPY 4,255,128,2
0040 END
```

## GDISP

### FONCTION

Cette commande permet d'afficher du texte dans la zone graphique afin d'annoter un graphique, par exemple.

### FORME GÉNÉRALE

GDISP X, Y, EchX, EchY, C, Dir, Texte

### RÈGLES

- X et Y sont les coordonnées de début de l'affichage (bas gauche du texte). Les valeurs doivent être comprises entre 0 et 255.
- EchX et EchY sont les facteurs d'échelle dans le sens des abscisses (EchX) et des ordonnées (EchY). Leurs valeurs doivent être comprises entre 1 et 15.
- C est la couleur de tracé. Sa valeur doit être comprise entre 0 et 15.
- Dir est la direction du texte. Sa valeur doit être comprise entre 0 et 7 et correspond aux directions données par le tableau suivant :

Valeur de Dir	Direction géographique du tracé
0	Est
1	Sud-Est
2	Sud
3	Sud-Ouest
4	Ouest
5	Nord-Ouest
6	Nord
7	Nord-Est

- Les caractères sont tracés au moyen du sous-programme PLOT. La commande GPATTERN est donc active. Pour un résultat optimal, il est conseillé d'exécuter GPATTERN 65535 avant d'utiliser GDISP.
- Si le tracé déborde d'un côté de l'écran graphique, le tracé se poursuit dans la zone diamétralement opposée.
- L'absence d'un paramètre ou une valeur erronée déclenchent une erreur 98.
- Tous les caractères affichables en mode alphanumérique peuvent être affichés en mode graphique (sauf les caractères semi-graphiques, naturellement).



## EXEMPLES

```
0010 REM TRACE DANS HUIT DIRECTIONS
0020 DIM A$(10)
0030 XC=YC=127:REM COORDONNEES DU CENTRE
0040 RAYON=20
0050 FOR I=0 TO -315 STEP -45
0060 REM CALCUL DES COORDONNEES DU POINT D'AFFICHAGE
0070 REM SITUE SUR UN CERCLE DE RAYON 20
0080 XD=XC+RAYON*COS I:YD=YC+RAYON*SIN I
0090 READ A$
0100 GDISP XD,YD,1,1,1,-I/45,A$
0110 NEXT I
0120 DATA EST,SUD-EST,SUD,SUD-OUEST
0130 DATA OUEST,NORD-OUEST,NORD-NORD-EST
0140 END

0010 REM AFFICHAGE DU JEU DE CARACTERES (ECHELLE 2)
0020 DIM A$(128),B$(20)
0030 FOR I=16 TO 127:REM REMPLISSAGE A$ PAR CODES VALIDES
0040 A$(I-15)=BIN I
0050 NEXT I
0060 X=0:Y=235:REM COORDONNEES D'AFFICHAGE
0070 FOR I=1 TO 6
0080 B$=A$(1+20*(I-1);20*I):REM DIVISE A$ EN BLOCS DE 20
0090 GDISP X,Y,2,2,1,0,B$
0100 Y=Y-20:X=0:REM COORDONNEES LIGNE SUIVANTE
0110 NEXT I
0120 END
```